

```
log using "D:\Jason\workshop\Margins\2023\margins workshop 4.log", replace
set more 1

use http://www.stata-press.com/data/r14/margex, clear

sum y outcome sex age

*****
* 1. Continuous Dependent variables
*****

use http://www.stata-press.com/data/r14/margex, clear

reg y i.sex##c.age

*****
* 1.1 Average Adjusted Prediction and Average Marginal Effect
*****
margins i.sex
margins, dydx(i.sex)

*****
* 1.2 Adjusted Predictions and Marginal Effects at the Means
*****
margins i.sex, atmeans
margins, dydx(i.sex) atmeans

*****
* 1.3 Adjusted Predictions and Marginal Effects at Representative values
*****
margins i.sex, at(age=(20(10)60))
margins, dydx(i.sex) at(age=(20(10)60))

*****
* 2. Binary Dependent variables
*****

logit outcome i.sex##c.age

*****
* 2.1 Average Adjusted Prediction and Average Marginal Effect
*****
margins i.sex
margins, dydx(i.sex)

*****
* 2.2 Adjusted Predictions and Marginal Effects at the Means
*****
margins i.sex, atmeans
margins, dydx(i.sex) atmeans

*****
* 2.3 Adjusted Predictions and Marginal Effects at Representative values
*****
margins i.sex, at(age=(20(10)60))
margins, dydx(i.sex) at(age=(20(10)60))
```

```
*****  
* 3. Nominal Dependent Variable  
*****
```

```
mlogit group i.sex##c.age
```

```
*****  
* 3.1 Average Adjusted Prediction and Average Marginal Effect  
*****  
margins i.sex  
margins, dydx(i.sex)
```

```
*****  
* 3.2 Adjusted Predictions and Marginal Effects at the Means  
*****  
margins i.sex, atmeans  
margins, dydx(i.sex) atmeans
```

```
*****  
* 3.3 Adjusted Predictions and Marginal Effects at Representative values  
*****  
margins i.sex, at(age=(20(10)60))  
margins, dydx(i.sex) at(age=(20(10)60))
```

```
*****  
* 4. Data collected with complex survey design  
*****  
use "D:\Jason\workshop\Margins\2023\margins.dta", clear
```

```
svydes
```

```
sum illness sex age
```

```
svy: reg illness i.sex##c.age
```

```
*****  
* 4.1 Average Adjusted Prediction and Average Marginal Effect  
*****  
margins i.sex, vce(unconditional)  
margins, dydx(i.sex) vce(unconditional)
```

```
*****  
* 4.2 Adjusted Predictions and Marginal Effects at the Means  
*****  
margins i.sex, atmeans vce(unconditional)  
margins, dydx(i.sex) atmeans vce(unconditional)
```

```
*****  
* 4.3 Adjusted Predictions and Marginal Effects at Representative values  
*****  
margins i.sex, at(age=(20(10)80)) vce(unconditional)  
margins, dydx(i.sex) at(age=(20(10)80)) vce(unconditional)
```

```
*****  
* 5. Margins with the imputed data
```

```
webuse mheart1s20, clear
mi convert flong
mi estimate , saving(D:\Jason\workshop\Margins\2023\miestfile.dta, replace) esample(esample) : logit attack smokes age bmi hsgrad female
```

* 5.1 Average Adjusted Prediction and Average Marginal Effect

```
mimrgns using D:\Jason\workshop\Margins\2023\miestfile.dta , esample(esample) predict(pr) dydx(*)
mimrgns using D:\Jason\workshop\Margins\2023\miestfile.dta , esample(esample) dydx(*)
```

* 6. Plotting the results from the -margins- command

```
use http://www.stata-press.com/data/r14/margex, clear
mlogit group i.sex##c.age
```

* Plotting the Adjusted Predictions

```
margins i.sex, at(age=(20(10)60))
marginsplot, yline(0)
graph save D:\Jason\workshop\Margins\2023\graph1.gph, replace
```

```
margins i.sex, at(age=(20(10)60))
marginsplot, by(sex) yline(0)
graph save D:\Jason\workshop\Margins\2023\graph2.gph, replace
```

```
log close
```